[bookmark: _GoBack]REPORT  zfsl_spj_psem.

************************************************************************
*                              TABLES                                  *
************************************************************************
TABLES: pernr.

************************************************************************
*                            INFOTYPES                                 *
************************************************************************
INFOTYPES:  0000, " Actions
            0001, " Orgnizational Assignment
            0009. " Bank Details

************************************************************************
*                              NODES                                   *
************************************************************************
NODES: payroll TYPE pay99_result.

************************************************************************
*                          Types Defined                               *
************************************************************************
TYPES:  BEGIN OF ty_pernr,
        pernr TYPE persno,
        ename TYPE emnam,
        plans TYPE plans,
        stext TYPE stext,
        persk TYPE persk,
        ptext TYPE pktxt,
        werks TYPE persa,
        btrtl TYPE btrtl,
        btext TYPE btrtx,
        persg  TYPE persg,
        pgtxt  TYPE pgtxt,
        orgeh TYPE orgeh,
        orgtx TYPE orgtx,
        bankl  TYPE bankk,
        bankn  TYPE bankn,
        bkort  TYPE ort01,
        email TYPE comm_id_long,
        mobno TYPE comm_id_long,
        END OF ty_pernr.

TYPES: BEGIN OF ty_hrpy_rgdir.
        INCLUDE TYPE hrpy_rgdir.
TYPES: END OF ty_hrpy_rgdir.

TYPES:  BEGIN OF ty_lgart,
        lgart   TYPE lgart,
        lgtxt   TYPE lgtxt,
        END OF ty_lgart.
TYPES:  BEGIN OF ty_sdata,
        pernr TYPE persno,
        sortk TYPE char1,
        lgart TYPE lgart,
        lgtxt TYPE lgtxt,
        betrg TYPE betrg,
        END OF ty_sdata.

************************************************************************
*             Declaration of Variables & Internal Tables               *
************************************************************************
DATA: it_pernr          TYPE STANDARD TABLE OF ty_pernr,
      wa_pernr          TYPE ty_pernr,
      it_sdata          TYPE STANDARD TABLE OF ty_sdata,
      wa_sdata          TYPE ty_sdata.
DATA: it_hrpy_rgdir     TYPE STANDARD TABLE OF ty_hrpy_rgdir,
      wa_hrpy_rgdir     TYPE ty_hrpy_rgdir.
DATA: it_lgart          TYPE SORTED TABLE OF ty_lgart WITH UNIQUE KEY lgart,
      wa_lgart          TYPE ty_lgart.
DATA: it_payroll_result TYPE pay99_result.
DATA: wa_rt             LIKE LINE OF payroll-inter-rt.
DATA: it_lines          TYPE i.
DATA: fpper             TYPE faper.
DATA: cmn               TYPE p33_regno. "t7ru9a-regno.
DATA: cmy               TYPE string.
DATA: tabix             TYPE sytabix.
DATA: directory         TYPE string,
      rc                TYPE i.
DATA: save_path         TYPE string.

************************************************************************
*                             RANGEs                                   *
************************************************************************
DATA: it_rg_lgart       TYPE RANGE OF lgart,
      wa_rg_lgart       LIKE LINE OF it_rg_lgart.

************************************************************************
*                         FIELD-SYMBOLS                                *
************************************************************************
FIELD-SYMBOLS: <wa_sdata> TYPE ty_sdata.

************************************************************************
*                              INCLUDEs                                *
************************************************************************
INCLUDE:  ole2incl, excel__c.

************************************************************************
*                          INITIALIZATION                              *
************************************************************************
INITIALIZATION.
  PERFORM initialization.

************************************************************************
*                        START-OF-SELECTION                            *
************************************************************************
START-OF-SELECTION.
  PERFORM start_of_selection.

************************************************************************
*                              GET pernr                               *
************************************************************************
GET pernr.
  PERFORM get_pernr.

************************************************************************
*                         END-OF-SELECTION                             *
************************************************************************
END-OF-SELECTION.
  PERFORM end_of_selection.



*&---------------------------------------------------------------------*
*&      Form  INITIALIZATION
*&---------------------------------------------------------------------*
*       text
*----------------------------------------------------------------------*
FORM initialization.

  IF it_rg_lgart[] IS INITIAL.
**** I am using few Standard Wage types only
**** You can add more as per your System
    wa_rg_lgart-sign = 'I'.
    wa_rg_lgart-option = 'EQ'.
    wa_rg_lgart-low = '/560'. APPEND wa_rg_lgart TO it_rg_lgart. " Net Payment
    wa_rg_lgart-low = '/101'. APPEND wa_rg_lgart TO it_rg_lgart. " Gross
    wa_rg_lgart-low = '/110'. APPEND wa_rg_lgart TO it_rg_lgart. " Net Dedications
    wa_rg_lgart-low = '0001'. APPEND wa_rg_lgart TO it_rg_lgart. " Basic
    wa_rg_lgart-low = '0002'. APPEND wa_rg_lgart TO it_rg_lgart. " House Rent
    wa_rg_lgart-low = '0006'. APPEND wa_rg_lgart TO it_rg_lgart. " Utility Allowance
    wa_rg_lgart-low = '6000'. APPEND wa_rg_lgart TO it_rg_lgart. " Income Tax
    wa_rg_lgart-low = '3001'. APPEND wa_rg_lgart TO it_rg_lgart. " EOBI
    wa_rg_lgart-low = '3003'. APPEND wa_rg_lgart TO it_rg_lgart. " PF
    SORT it_rg_lgart BY low.
    DELETE ADJACENT DUPLICATES FROM it_rg_lgart COMPARING low.

    SELECT lgart lgtxt
      INTO CORRESPONDING FIELDS OF TABLE it_lgart
      FROM t512t
      WHERE sprsl = 'EN'
        AND molga = '99'
        AND lgart IN it_rg_lgart.
  ENDIF.

ENDFORM.                    "INITIALIZATION

*&---------------------------------------------------------------------*
*&      Form  start_of_selection
*&---------------------------------------------------------------------*
*       text
*----------------------------------------------------------------------*
FORM start_of_selection.

  IF pnppabrj IS NOT INITIAL.
    fpper+0(4) = pnppabrj.
    fpper+4(2) = pnppabrp.
  ENDIF.

  IF pnpdispj IS NOT INITIAL.
    fpper+0(4) = pnpdispj.
    fpper+4(2) = pnpdispp.
  ENDIF.

  CALL FUNCTION 'HR_RU_MONTH_NAME_IN_GENITIVE'
    EXPORTING
      month = pn-endda+4(2)
    IMPORTING
      name  = cmn.

  TRANSLATE cmn(1) TO UPPER CASE.
  CONCATENATE cmn '-' pn-endda+0(4) INTO cmy SEPARATED BY space.

ENDFORM.                    "start_of_selection

*&---------------------------------------------------------------------*
*&      Form  get_pernr
*&---------------------------------------------------------------------*
*       text
*----------------------------------------------------------------------*
FORM get_pernr.

  CLEAR: p0000, p0001, p0009.
  rp_provide_from_last p0000 space pn-begda '99991231'.
  rp_provide_from_last p0001 space pn-begda '99991231'.
  rp_provide_from_last p0009 space pn-begda '99991231'.

  wa_pernr-pernr = p0000-pernr.
  wa_pernr-ename = p0001-ename.
  wa_pernr-plans = p0001-plans.
  wa_pernr-persk = p0001-persk.
  wa_pernr-werks = p0001-werks.
  wa_pernr-btrtl = p0001-btrtl.
  wa_pernr-persg = p0001-persg.
  wa_pernr-orgeh = p0001-orgeh.
  wa_pernr-bankl = p0009-bankl.
  wa_pernr-bankn = p0009-bankn.
  wa_pernr-bkort = p0009-bkort.

  SELECT SINGLE stext
    INTO wa_pernr-stext
    FROM hrp1000
    WHERE plvar = '01'
      AND otype = 'S'
      AND objid = wa_pernr-plans
      AND istat = '1'.

  SELECT SINGLE ptext
    INTO wa_pernr-ptext
    FROM t503t
    WHERE sprsl = 'E'
      AND persk = wa_pernr-persk.

  SELECT SINGLE btext
    INTO wa_pernr-btext
    FROM t001p
    WHERE werks = wa_pernr-werks
      AND btrtl = wa_pernr-btrtl.

  SELECT SINGLE ptext
    INTO wa_pernr-pgtxt
    FROM t501t
    WHERE sprsl = 'EN'
      AND persg = wa_pernr-persg.

  SELECT SINGLE orgtx
    INTO wa_pernr-orgtx
    FROM t527x
    WHERE sprsl = 'EN'
      AND orgeh = wa_pernr-orgeh
      AND endda = '99991231'.

  SELECT SINGLE usrid
    INTO wa_pernr-email
    FROM pa0105
    WHERE pernr = wa_pernr-pernr
      AND subty = '0005'. " Subtype for E-Mail can be change in your System, Change accordingly

  SELECT SINGLE usrid
    INTO wa_pernr-mobno
    FROM pa0105
    WHERE pernr = wa_pernr-pernr
      AND subty = '0004'. " Subtype for Mobile can be change in your System, Change accordingly

  APPEND wa_pernr TO it_pernr.

ENDFORM.                    "get_pernr

*&---------------------------------------------------------------------*
*&      Form  end_of_selection
*&---------------------------------------------------------------------*
*       text
*----------------------------------------------------------------------*
FORM end_of_selection.

  PERFORM:  get_and_arrange_payroll_data.
  PERFORM:  create_directory.
  PERFORM:  generate_payslip_excel.

ENDFORM.                    "end_of_selection

*&---------------------------------------------------------------------*
*&      Form  get_and_arrange_payroll_data
*&---------------------------------------------------------------------*
*       text
*----------------------------------------------------------------------*
FORM get_and_arrange_payroll_data.

  SORT: it_pernr BY pernr.
  DELETE ADJACENT DUPLICATES FROM it_pernr[] COMPARING pernr.

*  Get all payrolls
  SELECT * INTO CORRESPONDING FIELDS OF TABLE it_hrpy_rgdir
    FROM hrpy_rgdir
    FOR ALL ENTRIES IN it_pernr
    WHERE pernr = it_pernr-pernr
      AND srtza = 'A'               "Only current results.
      AND fpper = fpper
      AND abkrs = pnpxabkr.

  SORT it_hrpy_rgdir BY pernr fpbeg.

**  Append data of all Wage Types
  LOOP AT it_pernr INTO wa_pernr.
    LOOP AT it_hrpy_rgdir INTO wa_hrpy_rgdir WHERE pernr = wa_pernr-pernr.
      PERFORM get_rt USING wa_hrpy_rgdir-pernr wa_hrpy_rgdir-seqnr.
      IF sy-subrc = 0.
        DESCRIBE TABLE it_payroll_result-inter-rt LINES it_lines.
        IF it_lines = 0.
          CONTINUE.
        ENDIF.
        LOOP AT it_payroll_result-inter-rt INTO wa_rt WHERE lgart IN it_rg_lgart.
          wa_sdata-pernr = wa_pernr-pernr.
          wa_sdata-lgart = wa_rt-lgart.
          wa_sdata-betrg = wa_rt-betrg.
          READ TABLE it_lgart INTO wa_lgart WITH TABLE KEY lgart = wa_sdata-lgart.
          IF sy-subrc = 0.
            wa_sdata-lgtxt = wa_lgart-lgtxt.
          ENDIF.
          COLLECT wa_sdata INTO it_sdata.
        ENDLOOP.
      ENDIF.
    ENDLOOP.
  ENDLOOP.

**  Delete pernr if not Payroll data
  LOOP AT it_pernr INTO wa_pernr.
    tabix = sy-tabix.
    READ TABLE it_sdata INTO wa_sdata WITH KEY pernr = wa_pernr-pernr.
    IF sy-subrc <> 0.
      DELETE it_pernr INDEX tabix.
    ENDIF.
  ENDLOOP.

**  Delete if have 0(Zero) Value for any wage type
  DELETE it_sdata WHERE betrg IS INITIAL.

**  Arrange wage types
  LOOP AT it_sdata ASSIGNING <wa_sdata>.
    IF <wa_sdata>-betrg > 0.
      <wa_sdata>-sortk = '1'.
    ENDIF.
    IF <wa_sdata>-lgart = '/101'.
      <wa_sdata>-sortk = '2'.
    ENDIF.
    IF <wa_sdata>-betrg < 0.
      <wa_sdata>-sortk = '3'.
    ENDIF.
    IF <wa_sdata>-lgart = '/110'.
      <wa_sdata>-sortk = '4'.
    ENDIF.
    IF <wa_sdata>-lgart = '/560'.
      <wa_sdata>-sortk = '5'.
    ENDIF.
  ENDLOOP.

**  Sorting
  SORT: it_sdata BY pernr sortk lgtxt.

ENDFORM.                    "get_and_arrange_payroll_data

*&---------------------------------------------------------------------*
*&      Form  get_rt
*&---------------------------------------------------------------------*
*       text
*----------------------------------------------------------------------*
*      -->PERNR      text
*      -->SEQNR      text
*----------------------------------------------------------------------*
FORM get_rt USING pernr seqnr.
  CLEAR: it_payroll_result.
  CALL FUNCTION 'PYXX_READ_PAYROLL_RESULT'
    EXPORTING
      clusterid                    = 'RX'
      employeenumber               = pernr
      sequencenumber               = seqnr
    CHANGING
      payroll_result               = it_payroll_result
    EXCEPTIONS
      illegal_isocode_or_clusterid = 1
      error_generating_import      = 2
      import_mismatch_error        = 3
      subpool_dir_full             = 4
      no_read_authority            = 5
      no_record_found              = 6
      versions_do_not_match        = 7
      error_reading_archive        = 8
      error_reading_relid          = 9
      OTHERS                       = 10.
ENDFORM.                    "get_rt

*&---------------------------------------------------------------------*
*&      Form  create_directory
*&---------------------------------------------------------------------*
*       text
*----------------------------------------------------------------------*
FORM create_directory.

  CONCATENATE: 'C:\SAP HR Temp' '' INTO directory.

  IF it_pernr[] IS NOT INITIAL.
    CALL METHOD cl_gui_frontend_services=>directory_create
      EXPORTING
        directory                = directory
      CHANGING
        rc                       = rc
      EXCEPTIONS
        directory_create_failed  = 1
        cntl_error               = 2
        error_no_gui             = 3
        directory_access_denied  = 4
        directory_already_exists = 5
        path_not_found           = 6
        unknown_error            = 7
        not_supported_by_gui     = 8
        wrong_parameter          = 9
        OTHERS                   = 10.
    IF sy-subrc <> 0 AND sy-subrc <> 5.
      MESSAGE ID sy-msgid TYPE sy-msgty NUMBER sy-msgno
                 WITH sy-msgv1 sy-msgv2 sy-msgv3 sy-msgv4.
    ENDIF.

    CONCATENATE: 'C:\SAP HR Temp\' sy-datum INTO directory.
    CALL METHOD cl_gui_frontend_services=>directory_create
      EXPORTING
        directory                = directory
      CHANGING
        rc                       = rc
      EXCEPTIONS
        directory_create_failed  = 1
        cntl_error               = 2
        error_no_gui             = 3
        directory_access_denied  = 4
        directory_already_exists = 5
        path_not_found           = 6
        unknown_error            = 7
        not_supported_by_gui     = 8
        wrong_parameter          = 9
        OTHERS                   = 10.
    IF sy-subrc <> 0 AND sy-subrc <> 5.
      MESSAGE ID sy-msgid TYPE sy-msgty NUMBER sy-msgno
                 WITH sy-msgv1 sy-msgv2 sy-msgv3 sy-msgv4.
    ENDIF.
  ENDIF.

ENDFORM.                    "create_directory


*&---------------------------------------------------------------------*
*&      Form  Generate_payslip_excel
*&---------------------------------------------------------------------*
*       text
*----------------------------------------------------------------------*
FORM generate_payslip_excel.
**  Generate Pay Slips with EXCEL

  DATA: obj_excel     TYPE ole2_object,
        obj_workbook  TYPE ole2_object,
        obj_worksheet TYPE ole2_object,
        obj_cell      TYPE ole2_object,
        obj_border    TYPE ole2_object,
        obj_font      TYPE ole2_object,
        obj_range     TYPE ole2_object.
  DATA: row_count     TYPE i,
        col_count_01  TYPE i,
        col_count_02  TYPE i.
  DATA: caddx         TYPE string,
        caddy         TYPE string.
  DATA: sortk         TYPE char1.


  LOOP AT it_pernr INTO wa_pernr.

    CREATE OBJECT obj_excel 'EXCEL.APPLICATION'.
    SET PROPERTY OF obj_excel  'VISIBLE' = 0.

    CALL METHOD OF
        obj_excel
        'WORKBOOKS' = obj_workbook.

    CALL METHOD OF
        obj_workbook
        'ADD'.

    CALL METHOD OF
        obj_excel
        'WORKSHEETS' = obj_worksheet
      EXPORTING
        #1           = 1.

    CALL METHOD OF
        obj_worksheet
        'ACTIVATE'.
    SET PROPERTY OF obj_worksheet
    'NAME' = 'Pay Slip'.

    CALL METHOD OF
        obj_excel
        'Range'   = obj_range
      EXPORTING
        #1        = 'A2'
        #2        = 'I2'.

    CALL METHOD OF
        obj_range
        'MERGE'.

    CALL METHOD OF
        obj_range
        'FONT'    = obj_font.

    SET PROPERTY OF obj_font 'BOLD' = 1.
    SET PROPERTY OF obj_font 'SIZE' = 15.

    CALL METHOD OF
        obj_excel
        'Cells'   = obj_cell
      EXPORTING
        #1        = 2
        #2        = 1.
    SET PROPERTY OF obj_cell 'Value' = 'XYZ (Company Name)' .
    SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlcenter.

    CALL METHOD OF
        obj_excel
        'Range'   = obj_range
      EXPORTING
        #1        = 'A3'
        #2        = 'I3'.

    CALL METHOD OF
        obj_range
        'MERGE'.

    CALL METHOD OF
        obj_range
        'FONT'    = obj_font.

    SET PROPERTY OF obj_font 'BOLD' = 1.
    SET PROPERTY OF obj_font 'SIZE' = 15.

    CALL METHOD OF
        obj_excel
        'Cells'   = obj_cell
      EXPORTING
        #1        = 3
        #2        = 1.
    CONCATENATE: 'Pay Slip for the month of' cmy INTO cmy SEPARATED BY space.
    SET PROPERTY OF obj_cell 'Value' = cmy.
    SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlcenter.

    CALL METHOD OF
        obj_excel
        'Cells'   = obj_cell
      EXPORTING
        #1        = 5
        #2        = 4.
    SET PROPERTY OF obj_cell 'Value' = 'Employee Number:' .
    SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlright.

    CALL METHOD OF
        obj_excel
        'Cells'   = obj_cell
      EXPORTING
        #1        = 5
        #2        = 6.
    SET PROPERTY OF obj_cell 'Value' = wa_pernr-pernr .
    SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlleft.

    CALL METHOD OF
        obj_excel
        'Cells'   = obj_cell
      EXPORTING
        #1        = 6
        #2        = 4.
    SET PROPERTY OF obj_cell 'Value' = 'Employee Name:' .
    SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlright.

    CALL METHOD OF
        obj_excel
        'Cells'   = obj_cell
      EXPORTING
        #1        = 6
        #2        = 6.
    SET PROPERTY OF obj_cell 'Value' = wa_pernr-ename .
    SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlleft.

    CALL METHOD OF
        obj_excel
        'Cells'   = obj_cell
      EXPORTING
        #1        = 7
        #2        = 4.
    SET PROPERTY OF obj_cell 'Value' = 'Designation:' .
    SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlright.

    CALL METHOD OF
        obj_excel
        'Cells'   = obj_cell
      EXPORTING
        #1        = 7
        #2        = 6.
    SET PROPERTY OF obj_cell 'Value' = wa_pernr-stext.
    SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlleft.

    CALL METHOD OF
        obj_excel
        'Cells'   = obj_cell
      EXPORTING
        #1        = 8
        #2        = 4.
    SET PROPERTY OF obj_cell 'Value' = 'Department:' .
    SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlright.

    CALL METHOD OF
        obj_excel
        'Cells'   = obj_cell
      EXPORTING
        #1        = 8
        #2        = 6.
    SET PROPERTY OF obj_cell 'Value' = wa_pernr-orgtx .
    SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlleft.

    CALL METHOD OF
        obj_excel
        'Range'   = obj_range
      EXPORTING
        #1        = 'D5'
        #2        = 'F8'.

    CALL METHOD OF
        obj_range
        'FONT'    = obj_font.

    SET PROPERTY OF obj_font 'BOLD' = 1.
    SET PROPERTY OF obj_font 'SIZE' = 12.

    CALL METHOD OF
        obj_excel
        'Cells'   = obj_cell
      EXPORTING
        #1        = 10
        #2        = 1.
    SET PROPERTY OF obj_cell 'Value' = 'Recurring Payments' .
    SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlleft.
    CALL METHOD OF
        obj_cell
        'FONT'   = obj_font.
    SET PROPERTY OF obj_font 'BOLD' = 1.
    SET PROPERTY OF obj_font 'SIZE' = 12.

    CALL METHOD OF
        obj_excel
        'Cells'   = obj_cell
      EXPORTING
        #1        = 10
        #2        = 9.
    SET PROPERTY OF obj_cell 'Value' = 'Addintional Payments' .
    SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlright.
    CALL METHOD OF
        obj_cell
        'FONT'   = obj_font.
    SET PROPERTY OF obj_font 'BOLD' = 1.
    SET PROPERTY OF obj_font 'SIZE' = 12.

    row_count = 11.

    LOOP AT it_sdata INTO wa_sdata WHERE pernr = wa_pernr-pernr.
      IF wa_sdata-sortk <> sortk.
        IF wa_sdata-sortk = '2' OR wa_sdata-sortk = '4' OR wa_sdata-sortk = '5'.
          ADD 1 TO row_count.
          caddx = caddy = row_count.
          CONCATENATE:  'A' caddx INTO caddx,
                        'I' caddy INTO caddy.
          CONDENSE: caddx,
                    caddy.

          CALL METHOD OF
              obj_excel
              'Range'   = obj_range
            EXPORTING
              #1        = caddx
              #2        = caddy.
          CALL METHOD OF
              obj_range
              'BORDERS' = obj_border
            EXPORTING
              #1        = '3'.
          SET PROPERTY OF obj_border 'LineStyle' = '1'.
          SET PROPERTY OF obj_border 'WEIGHT' = 2.
          CALL METHOD OF
              obj_range
              'BORDERS' = obj_border
            EXPORTING
              #1        = '4'.
          SET PROPERTY OF obj_border 'LineStyle' = '1'.
          SET PROPERTY OF obj_border 'WEIGHT' = 2.
          FREE OBJECT obj_border.
        ENDIF.

      ENDIF.

      IF wa_sdata-sortk = '1' OR wa_sdata-sortk = '3'.
        col_count_01 = 1.
        col_count_02 = 4.
      ELSEIF wa_sdata-sortk = '2' OR wa_sdata-sortk = '4' OR wa_sdata-sortk = '5'.
        col_count_01 = 5.
        col_count_02 = 8.
      ENDIF.

      CALL METHOD OF
          obj_excel
          'Cells'   = obj_cell
        EXPORTING
          #1        = row_count
          #2        = col_count_01.
      SET PROPERTY OF obj_cell 'Value' = wa_sdata-lgtxt.
      SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlleft.
      IF wa_sdata-sortk = '2' OR wa_sdata-sortk = '4' OR wa_sdata-sortk = '5'.
        CALL METHOD OF
            obj_cell
            'FONT'   = obj_font.
        SET PROPERTY OF obj_font 'BOLD' = 1.
        SET PROPERTY OF obj_font 'SIZE' = 11.
      ENDIF.

      CALL METHOD OF
          obj_excel
          'Cells'   = obj_cell
        EXPORTING
          #1        = row_count
          #2        = col_count_02.

      SET PROPERTY OF obj_cell 'Value' = wa_sdata-betrg.
      SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlright.
      SET PROPERTY OF obj_cell 'Style' = 'Comma'.
      SET PROPERTY OF obj_cell 'NumberFormat' = '#,##0.00'.
      IF wa_sdata-sortk = '2' OR wa_sdata-sortk = '4' OR wa_sdata-sortk = '5'.
        CALL METHOD OF
            obj_cell
            'FONT'   = obj_font.
        SET PROPERTY OF obj_font 'BOLD' = 1.
        SET PROPERTY OF obj_font 'SIZE' = 11.
      ENDIF.

      IF wa_sdata-sortk <> sortk.
        IF wa_sdata-sortk = '2'.
          ADD 2 TO row_count.
          CALL METHOD OF
              obj_excel
              'Cells'   = obj_cell
            EXPORTING
              #1        = row_count
              #2        = 1.
          SET PROPERTY OF obj_cell 'Value' = 'Recurring Deductions' .
          SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlleft.
          CALL METHOD OF
              obj_cell
              'FONT'   = obj_font.
          SET PROPERTY OF obj_font 'BOLD' = 1.
          SET PROPERTY OF obj_font 'SIZE' = 12.

          CALL METHOD OF
              obj_excel
              'Cells'   = obj_cell
            EXPORTING
              #1        = row_count
              #2        = 9.
          SET PROPERTY OF obj_cell 'Value' = 'Addintional Deductions' .
          SET PROPERTY OF obj_cell 'HorizontalAlignment' = xlright.
          CALL METHOD OF
              obj_cell
              'FONT'   = obj_font.
          SET PROPERTY OF obj_font 'BOLD' = 1.
          SET PROPERTY OF obj_font 'SIZE' = 12.
        ENDIF.
      ENDIF.

      ADD 1 TO row_count.
      sortk = wa_sdata-sortk.

    ENDLOOP.

    GET PROPERTY OF obj_excel
      'ActiveWorkbook' = obj_workbook.

    CLEAR: save_path.
    CONCATENATE directory '\' wa_pernr-pernr '.xlsx' INTO save_path.

    CALL METHOD OF
        obj_workbook
        'SAVEAS'

      EXPORTING
        #1           = save_path.

    CALL METHOD OF
        obj_excel
        'Quit'.

    FREE OBJECT:  obj_border,
                  obj_range,
                  obj_font,
                  obj_cell,
                  obj_worksheet,
                  obj_workbook,
                  obj_excel.

  ENDLOOP.
ENDFORM.                    "Generate_payslip_excel

